// Appendix8.2

log using Appendix8.2.log, replace
use "C:\Users\sbstjp\OneDrive - Cardiff University\anes_timeseries_2024_stata_20250219.dta" // ANES 2024 Time Series Study // Preliminary Release: Pre-Election Data February 19, 2025 version

// Create social justice scale
* Rename variables 
rename V241290x Edi 
rename V241372x Tgbathroom 
rename V241375x Tgsport 
rename V241412x Appprotestgaza 

* Delete missing values
foreach var in Edi Tgbathroom Tgsport Appprotestgaza {
    replace `var' = . if `var' < 0
    tabulate `var', missing
}

*Reverse coding so social justice values are high
foreach var in Appprotestgaza Tgbathroom Edi {
    qui sum `var'
    local max_value = r(max)
    gen r`var' = `max_value' + 1 - `var'
}

*Standardize items in the scale from 1-2 - this avoids 0, for reasons outlined in next step
foreach var in rEdi rTgbathroom Tgsport rAppprotestgaza {
    summarize `var'
    gen s`var' = 1 + (`var' - r(min)) / (r(max) - r(min))
}

*At this point, the scale has a Cronbach's alpha of 0.76. 

* Replace missing values with 0 for the specified variables - this is necessary as Stata doesn't add up missing values and means a 0-1 standardization scale isn't feasible as missing values would overlap with the scale
foreach var in srEdi srTgbathroom sTgsport srAppprotestgaza {
    replace `var' = 0 if missing(`var')
}

* Initialize the total score and the count of non-zero responses
gen total_scoreSJV = 0
gen count_nonzeroSJV = 0

* Add each variable to the total scale score and count it if non-zero
foreach var in srEdi srTgbathroom sTgsport srAppprotestgaza  {
    replace total_scoreSJV = total_scoreSJV + `var'
    replace count_nonzeroSJV = count_nonzeroSJV + (`var' != 0)
}

* Calculate the average score, avoiding division by zero
gen SocJusValues = .
replace SocJusValues = total_scoreSJV / count_nonzeroSJV if count_nonzeroSJV > 0

// Liberalism scale
* Rename
rename V241269x bordersecurity
rename V241302 abortion
rename V241272x crime
rename V241308x deathpenalty
rename V241395x wall
rename V241389x birthrightcit
rename V241381x gayadoption
rename V241258 environment

*Drop missing values
foreach var in gayadoption bordersecurity crime deathpenalty wall birthrightcit {
    replace `var' = . if inlist(`var', -1, -2)
}

replace abortion = . if inlist(abortion, -9, -8, -1, 5)   
replace environment = . if inlist(environment, -9, -8, -1, 99)  

*Reverse code so liberal values are high
foreach var in gayadoption environment {
    qui sum `var'
    local max_value = r(max)
    gen r`var' = `max_value' + 1 - `var'
}
 
*Standardize items in the scale from 1-2 - this avoids 0, for reasons outlined in next step
foreach var in bordersecurity abortion crime deathpenalty wall birthrightcit rgayadoption renvironment {
    summarize `var'
    gen s`var' = 1 + (`var' - r(min)) / (r(max) - r(min))
}

* Replace missing values with 0 for the specified variables - this is necessary as Stata doesn't add up missing values and means a 0-1 standardization scale isn't feasible as missing values would overlap with the scale
foreach var in sbordersecurity sabortion scrime sdeathpenalty swall sbirthrightcit srgayadoption srenvironment {
    replace `var' = 0 if missing(`var')
}

* Initialize the total score and the count of non-zero responses
gen total_scoreAL = 0
gen count_nonzeroAL = 0

* Add each variable to the total scale score and count it if non-zero
foreach var in sbordersecurity sabortion scrime sdeathpenalty swall sbirthrightcit srgayadoption srenvironment {
    replace total_scoreAL = total_scoreAL + `var'
    replace count_nonzeroAL = count_nonzeroAL + (`var' != 0)
}

* Calculate the average score, avoiding division by zero
gen LibValues = .
replace LibValues = total_scoreAL / count_nonzeroAL if count_nonzeroAL > 0

// Demographics
*Delete missing values and rename
rename V241458x age
replace age=. if age<0 

rename V241566x income
replace income=. if income<0 

replace V241177=. if V241177==99 
replace V241177=. if V241177<0
rename V241177 libconsp

* Generate dummies
gen FemaleGender=.
replace FemaleGender=1 if V241551==1
replace FemaleGender=2 if V241551==2

gen Graduate=.
replace Graduate=0 if inrange(V241465x, 1, 3)
replace Graduate=1 if inlist(V241465x, 4, 5)

gen BIPOC=.
replace BIPOC=0 if V241501x==1
replace BIPOC=1 if inrange(V241501x, 2, 6)

// Standardize 
egen Age = std(age)
egen Income = std(income)

// Democratic questions
replace V241324=. if V241324<0
replace V241325=. if V241325<0

*Standardize items in the scale from 1-2, so coefficients can be compared to others in book
foreach var in V241324 V241325 {
    summarize `var'
    gen s`var' = 1 + (`var' - r(min)) / (r(max) - r(min))
}

rename sV241324 NewsOrgs
rename sV241325 BranchesOfGov

// Regressions
regress NewsOrgs LibValues [pweight=V240105a], robust 
eststo
regress NewsOrgs LibValues Age BIPOC FemaleGender Graduate Income [pweight=V240105a], robust 
eststo
regress NewsOrgs SocJusValues [pweight=V240105a], robust 
eststo
regress NewsOrgs SocJusValues Age BIPOC FemaleGender Graduate Income [pweight=V240105a], robust  
eststo
regress NewsOrgs LibValues SocJusValues Age BIPOC FemaleGender Graduate Income [pweight=V240105a], robust 
eststo
esttab

eststo clear

regress BranchesOfGov LibValues [pweight=V240105a], robust 
eststo
regress BranchesOfGov LibValues Age BIPOC FemaleGender Graduate Income [pweight=V240105a], robust 
eststo
regress BranchesOfGov SocJusValues [pweight=V240105a], robust 
eststo
regress BranchesOfGov SocJusValues Age BIPOC FemaleGender Graduate Income [pweight=V240105a], robust 
eststo
regress BranchesOfGov LibValues SocJusValues Age BIPOC FemaleGender Graduate Income [pweight=V240105a], robust 
eststo
esttab

log close

eststo clear

// Alternative specification
ologit NewsOrgs SocJusValues Age BIPOC FemaleGender Graduate Income [pweight=V240105a], robust // Alternative specification

